RESOLUTION DES EQUATIONS ALGEBRIQUES LINEAIRES

par Charles Hubert

Introduction

Pour résoudre un système d'équations linéaires, on peut penser que la primitive ⌹ résout la question. Dans certains problèmes on doit résoudre un système dont certains paramètres peuvent avoir plusieurs valeurs ; mais, au moins en APL*PLUS, ⌹ ne sait pas résoudre plusieurs systèmes à la fois :

```
      ⌹ 2 2 ⍴ 1 0 0 (1 2)
DOMAIN ERROR
      ⌹ 2 2 ⍴ 1 0 0 (1 2)
      ^
```

De plus ⌹ résout au sens des moindres carrés, même si l'argument droit est une matrice carrée, ce qui peut la mettre en difficulté quand certaines valeurs ont des ordres de grandeur très éloignés :

```
      ⌹ 2 2 ⍴ 1 0 0 1E¯20
DOMAIN ERROR
      ⌹ 2 2 ⍴ 1 0 0 1E¯20
      ^
```

Notons ici que la notion de moindres carrés n'a de sens que si toutes les équations s'expriment par des grandeurs de même nature : par exemple la somme des carrés de positions et de vitesses n'a pas de sens.

La fonction "EqaLin" présentée ici résout exactement les systèmes d'équations algébriques linéaires dont la matrice des coefficients est carrée.

La fonction "EqaLin"

Cette fonction évite les difficultés signalées dans l'introduction. Ordres de grandeurs éloignés :

```
⍝     (2 2 ⍴ 1 0 0 1) ⌹ 2 2 ⍴ 1 0 0 1E¯20
⍝
⍝
⍝
⍝
      EqaLin (2 2 ⍴ 1 0 0 1E¯20), 2 2 ⍴ 1 0 0 1
 1 0.0E0
 0 1.0E20
```

Systèmes multiples :

```
      '|','⍕'EqaLin(2 2 ⍴ 1 0 0 (1 2)), 2 2 ⍴ 1 0 0 1
 |1 1 |0 0
 |0 0 |1 0.5
```

Pour résoudre le système d'équations

```
       A X = B
```

où A est une matrice carrée on exécute

```
      X ← EqaLin A,B
```

et le résultat vérifie

```
      B = ⊃ A +.× X
```

B peut avoir plus d'une colonne.

Le rang de A peut être supérieur à 2, celui de B doit suivre ; par exemple si la dimension de A est (2,3,4,4), alors EqaLin résout 2×3 systèmes de 4 équations à 4 inconnues en même temps.

```
      ⍴x←EqaLin(a←÷1+a⍴⍳×/a← 2 3 4 4), 2 3 4 ⍴3+⍳24
2 3 4 1
```

Vérification :

```
      ⊃,¨(⊂[2 3]a)+.×¨⊂[2 3]x
  3  4  5  6
  7  8  9 10
 11 12 13 14

 15 16 17 18
 19 20 21 22
 23 24 25 26
```

EqaLin accepte les tableaux gigognes : certaines cases de A ou B peuvent contenir des tableaux :

```
⍝                           4 cas
⍝                         ┌───────┐
      a← 3 3 ρ 27 0 41 (1 2 3 4) 2 70 93 23 18
```

alors EqaLin résout 4 systèmes de 3 équations à 3 inconnues à la fois :

```
⍝                         4 cas
⍝                       ┌───────────┐
      ρx←EqaLin a, (51 11 41 58) 87 44
3 1
Vérification :
      ⊃⊃a+.×,x
 51 11 41 58
 87 87 87 87
 44 44 44 44
```

La fonction

On peut la trouver dans le fichier APL*PLUS

EQUAT.SF

La composante 1 est une table des matières, les autres composantes sont les ⎕vr des fonctions de ce fichier.

```
    ∇  x←EqaLin a;⎕io;⎕elx;ns;c;nc;l;nl;l0;l1;m;p;r
[1]     ⎕io←0 ⋄ ⎕elx←'⎕error(⎕dm⍳⎕tcnl)ρ⎕dm'
[2]     a←⊂[r←¯2↓⍳ρρa]a
[3]     →(1⊃ρx← 0 (0⊃ρa)↓a←(¯2↑ 1 1 ,ρa)ρa++/,a-a)↓C
[4]     a←⊃[0]∊¨a
[5]     (ns nl nc)←ρa
[6]       a←a×[0 1]0.5*⌊2⍟+/1E¯300,| ns nl nl ↑a
[7]     a← (ns×nl) nc ρa
[8]     l1← ns nl ρ⍳0⊃ρa ⋄ l0←l1[; 0 0] ⋄ c←⍳nc ⋄ l←¯1 ⋄ →B
[9]    A:p←l+p⍳¨m←⌈/¨0,¨p←⊂[1]|a[0 1 ↓l1;l]
[10]    ⎕error(0∊m)/'MATRICE SINGULIERE'
[11]    m←l0+l,[0.5]p ⋄ ((,l1)[m])←(,l1)[⌽m]
[12]    a[l1[;l];c]←÷/a[l1[;l];c,[0.5]l]
[13]    a[l1;c]←a[l1;c]-a[l1[;nlρl];c]×a[l1;(ρc)ρl]×[1]l≠⍳nl
[14]    c←1↓c
[15]   B:→(nl>l←l+1)↑A
[16]    (∊x)←, 2 0 1 ⍉a[l1;c]
[17]   C:x←⊃[r]x
    ∇
```